

# NAG Fortran Library Routine Document

## G08AKF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

G08AKF calculates the exact tail probability for the Mann–Whitney rank sum test statistic for the case where there are ties in the two samples pooled together.

### 2 Specification

```
SUBROUTINE G08AKF(N1, N2, TAIL, RANKS, U, P, WRK, LWRK, IWRK, IFAIL)
INTEGER          N1, N2, LWRK, IWRK(2*(N1+N2+1)), IFAIL
real           RANKS(N1+N2), U, P, WRK(LWRK)
CHARACTER*1      TAIL
```

### 3 Description

G08AKF computes the exact tail probability for the Mann–Whitney  $U$  test statistic (calculated by G08AHF and returned through the parameter  $U$ ) using a method based on an algorithm developed by Neumann (1988), for the case where there are ties in the pooled sample.

The Mann–Whitney  $U$  test investigates the difference between two populations defined by the distribution functions  $F(x)$  and  $G(y)$  respectively. The data consist of two independent samples of size  $n_1$  and  $n_2$ , denoted by  $x_1, x_2, \dots, x_{n_1}$  and  $y_1, y_2, \dots, y_{n_2}$ , taken from the two populations.

The hypothesis under test,  $H_0$ , often called the null hypothesis, is that the two distributions are the same, that is  $F(x) = G(x)$ , and this is to be tested against an alternative hypothesis  $H_1$  which is

$$H_1 : F(x) \neq G(y); \text{ or}$$

$$H_1 : F(x) < G(y), \text{ i.e., the } x\text{'s tend to be greater than the } y\text{'s; or}$$

$$H_1 : F(x) > G(y), \text{ i.e., the } x\text{'s tend to be less than the } y\text{'s,}$$

using a two-tailed, upper-tailed or lower-tailed probability respectively. The user selects the alternative hypothesis by choosing the appropriate tail probability to be computed (see the description of argument TAIL in Section 5).

Note that when using this test to test for differences in the distributions one is primarily detecting differences in the location of the two distributions. That is to say, if we reject the null hypothesis  $H_0$  in favour of the alternative hypothesis  $H_1: F(x) > G(y)$  we have evidence to suggest that the location, of the distribution defined by  $F(x)$ , is less than the location of the distribution defined by  $G(y)$ .

G08AKF returns the exact tail probability,  $p$ , corresponding to  $U$ , depending on the choice of alternative hypothesis,  $H_1$ .

The value of  $p$  can be used to perform a significance test on the null hypothesis  $H_0$  against the alternative hypothesis  $H_1$ . Let  $\alpha$  be the size of the significance test (that is  $\alpha$  is the probability of rejecting  $H_0$  when  $H_0$  is true). If  $p < \alpha$  then the null hypothesis is rejected. Typically  $\alpha$  might be 0.05 or 0.01.

### 4 References

Conover W J (1980) *Practical Nonparametric Statistics* Wiley

Neumann N (1988) Some procedures for calculating the distributions of elementary nonparametric test statistics *Statistical Software Newsletter* **14** (3) 120–126

Siegel S (1956) *Nonparametric Statistics for the Behavioral Sciences* McGraw-Hill

## 5 Parameters

- 1: N1 – INTEGER *Input*  
*On entry:* the number of non-tied pairs,  $n_1$ .  
*Constraint:*  $N1 \geq 1$ .
  
- 2: N2 – INTEGER *Input*  
*On entry:* the size of the second sample,  $n_2$ .  
*Constraint:*  $N2 \geq 1$ .
  
- 3: TAIL – CHARACTER\*1 *Input*  
*On entry:* indicates the choice of tail probability, and hence the alternative hypothesis.  
 If TAIL = 'T', then a two-tailed probability is calculated and the alternative hypothesis is  $H_1 : F(x) \neq G(y)$ .  
 If TAIL = 'U', then an upper-tailed probability is calculated and the alternative hypothesis is  $H_1 : F(x) < G(y)$ , i.e., the  $x$ 's tend to be greater than the  $y$ 's.  
 If TAIL = 'L', then a lower-tailed probability is calculated and the alternative hypothesis is  $H_1 : F(x) > G(y)$ , i.e., the  $x$ 's tend to be less than the  $y$ 's.  
*Constraint:* TAIL = 'T', 'U' or 'L'.
  
- 4: RANKS(N1+N2) – *real* array *Input*  
*On entry:* the ranks of the pooled sample. These ranks are output in the array RANKS by G08AHF and should not be altered in any way if the user is using the same  $n_1$ ,  $n_2$  and U as used in G08AHF.
  
- 5: U – *real* *Input*  
*On entry:* the value of the Mann–Whitney rank sum test statistic,  $U$ . This is the statistic returned through the parameter U by G08AHF.
  
- 6: P – *real* *Output*  
*On exit:* the tail probability,  $p$ , as specified by the parameter TAIL.
  
- 7: WRK(LWRK) – *real* array *Workspace*
- 8: LWRK – INTEGER *Input*  
*On entry:* the dimension of the array WRK as declared in the (sub)program from which G08AKF is called.  
  
*Constraint:*  $LWRK \geq n + n(n+1)(n+m) - \frac{n(n+1)(2 \times n+1)}{3} + 1$ , where  $n = \min(N1, N2)$  and  $m = \max(N1, N2)$ .
  
- 9: IWRK(2\*(N1+N2+1)) – INTEGER array *Workspace*
  
- 10: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by  $X04AAF$ ).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry,  $N1 < 1$ ,  
or  $N2 < 1$ .

$IFAIL = 2$

On entry,  $TAIL \neq 'T', 'U'$  or  $'L'$ .

$IFAIL = 3$

On entry,  $U < 0.0$ .

$IFAIL = 4$

On entry,  $LWRK$  is too small.

## 7 Accuracy

The exact tail probability,  $p$ , is computed to an accuracy of at least 4 significant figures.

## 8 Further Comments

The time taken by the routine increases with  $n_1$  and  $n_2$  and the product  $n_1 n_2$ . Note that the amount of workspace required becomes very large for even moderate sizes of  $n_1$  and  $n_2$ .

## 9 Example

The example program finds the Mann–Whitney test statistic, using  $G08AHF$  for two independent samples of size 16 and 23 respectively. This is used to test the null hypothesis that the distributions of the two populations from which the samples were taken are the same against the alternative hypothesis that the distributions are different. The test statistic, the approximate Normal statistic and the approximate two-tail probability are printed.  $G08AKF$  is then called to obtain the exact two-tailed probability. The exact probability is also printed.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G08AKF Example Program Text
*      Mark 14 Release.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MAXN1, MAXN2, MAXL, MAXIW
      PARAMETER        (MAXN1=25,MAXN2=25,MAXL=8000,MAXIW=100)
*      .. Local Scalars ..
      real             P, PEXACT, U, UNOR
      INTEGER          I, IFAIL, LWRK, N, N1, N2, NSUM
      LOGICAL          TIES
*      .. Local Arrays ..
      real             RANKS(MAXN1+MAXN2), WRK(MAXL), X(MAXN1), Y(MAXN2)
      INTEGER          IWRK(MAXIW)
*      .. External Subroutines ..
      EXTERNAL         G08AHF, G08AKF
*      .. Intrinsic Functions ..
```

```

      INTRINSIC          MIN
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G08AKF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N1, N2
      WRITE (NOUT,*)
      IF ((N1.LE.MAXN1) .AND. (N2.LE.MAXN2)) THEN
        WRITE (NOUT,99999) 'Sample size of group 1 = ', N1
        WRITE (NOUT,99999) 'Sample size of group 2 = ', N2
        WRITE (NOUT,*)
        READ (NIN,*) (X(I),I=1,N1)
        WRITE (NOUT,*) 'Mann-Whitney U test'
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Data values'
        WRITE (NOUT,*)
        WRITE (NOUT,99998) '      Group 1  ', (X(I),I=1,N1)
        READ (NIN,*) (Y(I),I=1,N2)
        WRITE (NOUT,*)
        WRITE (NOUT,99998) '      Group 2  ', (Y(I),I=1,N2)
        IFAIL = 0
*
      CALL G08AHF(N1,X,N2,Y,'Lower-tail',U,UNOR,P,TIES,RANKS,WRK,
+               IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99997) 'Test statistic      = ', U
      WRITE (NOUT,99997) 'Normal statistic   = ', UNOR
      WRITE (NOUT,99997) 'Tail probability   = ', P
      WRITE (NOUT,*)
      IF (TIES) THEN
        NSUM = N1 + N2
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Ranks'
        WRITE (NOUT,*)
        WRITE (NOUT,99998) '      Group 1  ', (RANKS(I),I=1,N1)
        WRITE (NOUT,*)
        WRITE (NOUT,99998) '      Group 2  ', (RANKS(I),I=N1+1,NSUM)
        N = MIN(N1,N2)
        LWRK = N + N*(N+1)*NSUM - N*(N+1)*(2*N+1)/3 + 1
        WRITE (NOUT,*)
        WRITE (NOUT,99996)
+       'The length of the workspace is calculated as ', LWRK
        IFAIL = 0
*
      CALL G08AKF(N1,N2,'Lower-tail',RANKS,U,PEXACT,WRK,LWRK,IWRK,
+               IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99997) 'Exact tail probability = ', PEXACT
      ELSE
        WRITE (NOUT,*)
+ 'There are no ties in the pooled sample so G08AKF was not called.'
      END IF
      ELSE
        WRITE (NOUT,*) 'Either N or M is out of range : '
        WRITE (NOUT,99995) 'N1 = ', N1, ' AND N2 = ', N2
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,A,8F5.1,2(/14X,8F5.1))
99997 FORMAT (1X,A,F10.4)
99996 FORMAT (1X,A,I10)
99995 FORMAT (1X,A,I16,A,I16)
      END

```

## 9.2 Program Data

G08AKF Example Program Data

```
16 23
13.0  6.0 12.0  7.0 12.0  7.0 10.0  7.0
10.0  7.0 16.0  7.0 10.0  8.0  9.0  8.0
17.0  6.0 10.0  8.0 15.0  8.0 15.0 10.0 15.0 10.0 14.0 10.0
14.0 11.0 14.0 11.0 13.0 12.0 13.0 12.0 13.0 12.0 12.0
```

## 9.3 Program Results

G08AKF Example Program Results

```
Sample size of group 1 =    16
Sample size of group 2 =    23
```

Mann-Whitney U test

Data values

```
Group 1  13.0  6.0 12.0  7.0 12.0  7.0 10.0  7.0
          10.0  7.0 16.0  7.0 10.0  8.0  9.0  8.0
```

```
Group 2  17.0  6.0 10.0  8.0 15.0  8.0 15.0 10.0
          15.0 10.0 14.0 10.0 14.0 11.0 14.0 11.0
          13.0 12.0 13.0 12.0 13.0 12.0 12.0
```

```
Test statistic      =    86.0000
Normal statistic    =   -2.8039
Tail probability     =    0.0025
```

Ranks

```
Group 1  29.5  1.5 24.5  5.0 24.5  5.0 16.0  5.0
          16.0  5.0 38.0  5.0 16.0  9.5 12.0  9.5
```

```
Group 2  39.0  1.5 16.0  9.5 36.0  9.5 36.0 16.0
          36.0 16.0 33.0 16.0 33.0 20.5 33.0 20.5
          29.5 24.5 29.5 24.5 29.5 24.5 24.5
```

The length of the workspace is calculated as       7633

Exact tail probability =       0.0020

---